

Confidence
ellipse
for (β_0, β_1)

```
DATA c; INFILE peru;
INPUT number age years weight height chin calf forearm pulse systol;

PROC MEANS NOPRINT;
VAR years; OUTPUT OUT=x n=n          /* sample size */
                        mean=xbar      /* mean of X_i */
                        css=Sxx;        /* sum of X_i**2 corrected for the mean (Sxx) */

PROC REG DATA=c OUTEST=est NOPRINT;
MODEL systol = years;

data est; set est;
  s = _RMSE;          /* root MSE = estimated standard deviation */
  b0 = INTERCEPT;   /* estimated intercept b0 */
  b1 = years;          /* estimated slope b1 */
KEEP s b0 b1;

DATA ellipse; MERGE x est;
  sb1 = s/sqrt(Sxx);   /* standard deviation of b1 */
  sb0 = s*sqrt(1/n+xbar**2/Sxx); /* standard deviation of b0 */
  F90 = 2*finv(0.90,2,n)*s**2; /* 90% upper bounds for the quadratic form */

do beta0=b0-3*sb0 by 0.1 to b0+3*sb0;
  D = (n*xbar*(beta0-b0))**2 - (n*xbar**2+Sxx)*(n*(beta0-b0)**2-F90); /* discriminant */
  if D < 0 then do; upper90 = .; lower90 = .; end;
  else do; upper90 = b1+(-n*xbar*(beta0-b0)+sqrt(D))/(n*xbar**2+Sxx);
        lower90 = b1+(-n*xbar*(beta0-b0)-sqrt(D))/(n*xbar**2+Sxx); end;
OUTPUT; END;

PROC PLOT DATA=ellipse;
PLOT (lower90 upper90)*beta0 = '' / overlay box;
RUN;
```

